
Abstract

Mathematical reasoning may involve several arithmetic types, including those
of the natural, integer, rational, real and complex numbers. These types satisfy
many of the same algebraic laws. These laws need to be made available to users,
uniformly and preferably without repetition, but with due account for the peculiari-
ties of each type. Subtyping, where a type inherits properties from a supertype, can
eliminate repetition only for a fixed type hierarchy set up in advance by implemen-
tors. The approach recently adopted for Isabelle usesaxiomatic type classes, an
established approach to overloading. Abstractions such as semirings, rings, fields
and their ordered counterparts are defined and theorems are proved algebraically.
Types that meet the abstractions inherit the appropriate theorems.

1 Introduction

Theorem-proving tools typically deal with many different types of numbers. The natu-
ral numbers are appropriate for foundational proofs relying on induction. The integers
and the rational numbers can be appropriate for modelling computer arithmetic. The
real and complex numbers are appropriate for developing a theory of mathematical
analysis. In the formalisation of mathematics, it is common to see several different
numeric types involved in a single theorem statement.

There are many different arithmetic laws. Addition and multiplication are typically
commutative and associative. Addition distributes over multiplication. Addition or
multiplication by a common value can be cancelled under certain circumstances. Other
laws relate to monotonicity and numerical signs. Still others concern subtraction, divi-
sion, absolute value, and so forth. A further complication is that a mathematical law,
when formalized in a theorem prover, often takes multiple forms. A mathematician
will apply a law such ask× (m+n) = k×m+k×n to a term of the form(m+n)×k.
Corollaries of a theorem—typically special cases where certain variables are replaced
by 0 or 1—give rise to additional theorems. The dozen or so laws in a textbook multiply
into hundreds of lemmas in a theorem prover.

With several numerical types and hundreds of arithmetic lemmas, we face a poten-
tial explosion: taking the obvious product will yield thousands of laws. The standard
method of avoiding this problem is subtyping. An example of subtyping is to formalize
the natural numbers as a subset of the reals, so that it inherits its laws from those of the
reals. Consider a slightly more complicated type hierarchy with the natural numbers,
the integers, the rationals, the reals and the complex numbers:

N ⊆ Z ⊆ Q ⊆ R ⊆ C

Subtyping in general is natural and useful, but using it to organise the arithmetic types
in a hierarchy has serious problems.

• It is inelegant, inverting the order of mathematical construction. Starting with the
natural numbers, we construct the integers, the rationals, the reals and finally the
complex numbers. The properties of the complex numbers are rightfully derived
from those of the natural numbers, not the other way round.

• It is inflexible. Only implementors can define the linear hierarchy to avoid the
repetition of arithmetic facts. If we develop a theory of complex polynomials,
then we would have to rewrite the standard prelude to make typecomplex a
subtype of typecomplexpoly . Of course, nobody would want to regard the
complex numbers as a subset of the complex polynomials.

• It is confusing. Users will expect the largest type to support the most operations.
Type complex includes all the other types as subsets, but it do not possess an
ordering. Properties involving orderings or signs, such as−a < −b ⇐⇒ b <

a, have to be inherited from typereal , while other properties are inherited from
typecomplex .

PVS uses subtyping to let theorems about the reals be re-used for the rationals, integers
and naturals, but it does not provide the complex numbers at all [11]. PVS provides an
abstract type, callednumber_field , which lies at the top of the arithmetic type hier-
archy. This type satisfies the axioms of a field, but not of an ordered field: “This allows
development of, for example, complex numbers or nonstandard reals as subtypes. . . .”

Abstract algebra offers a better approach to formalising arithmetic types. Roughly
speaking, afield is a set equipped with the operations of addition, subtraction, mul-
tiplication and division, satisfying all the familiar laws. The complex numbers are a
field. An ordered fieldis a field equipped with a linear ordering that is preserved in
the familiar way by addition and multiplication. The real and the rational numbers are
both ordered fields. Aring is similar to a field, but does not necessarily have division.
Precise definitions of these concepts appear below. Each one requires surprisingly few
axioms, and they build upon one another naturally, yielding a lattice of concepts. Given
such concepts, we can prove the familiar arithmetic laws starting from a minimal set of
axioms. Laws that hold for rings will continue to hold for ordered rings and for fields.
If we then define a type of complex polynomials and prove that it forms a ring, then it
immediately inherits all the properties proved for rings.

Most logical formalisms can easily express such an axiomatic development. Isa-
belle’s axiomatic type classesare particularly convenient [7, 13]. They provide con-
trolled polymorphism over a spectrum that ranges from traditional ML-style polymor-
phism to overloading. Concepts such as ring, ordered ring and field can be formalised
as axiomatic type classes. A type can be proved to belong to a type class, gaining
access to the theorems proved for that class. At same time, definitions of operators

can be specific to each type: addition has separate definitions for the natural numbers,
integers and rationals. Although the addition operator stands for a different function
for each arithmetic type, the abstract properties of addition are inherited according to
the axiomatic hierarchy.

This paper can be seen on two levels: first, as a suggestion for organising numerical
theories in general, and second, as an example of the use of axiomatic type classes in
Isabelle. The paper proceeds by introducing Isabelle and axiomatic type classes (§2). It
then describes the axiomatic type classes for arithmetic (§3). It shows how axiomatic
type classes support numeric literals (§4). It presents a small example—the natural
numbers extended with∞ (§5)—and finally concludes (§6).

2 Isabelle and Axiomatic Type Classes

Isabelle [8] is an interactive theorem prover implementing a higher order logical frame-
work. Isabelle supports a variety of formalisms; the most popular is higher-order logic
(Isabelle/HOL). Many large case studies have been done using Isabelle, concerning
for example cryptographic protocols [10] and the Java Virtual Machine [6]. For the
purposes of this paper, the most interesting aspect of Isabelle is its type system: order-
sorted polymorphism [7].

Polymorphism is found in a number of functional programming languages, such as
ML [9]. For example, the type of lists takes as an argument the type of the list elements.
The list reversal function ispolymorphic: it reverses any list without regard for the type
of the list elements. ML assignsrev the following type:

rev : (’a list) -> (’a list)

The symbol’a is a type variable, which ranges over all types. Here, we see thatrev

returns a list of the same type as it is given. Moreover, it executes the same algorithm (in
practice, the same code) without regard for the type of the list elements. Polymorphism
therefore differs fromoverloading, where a symbol such as + represents the concept of
addition for both integers and reals, but with separate addition algorithms for the two
types.

Polymorphism is as useful in logic as it is in programming. All versions of the
theorem prover HOL [4] use it, as does Isabelle. The formally defined list reversal
function is polymorphic, and so are the theorems proved about this function. For ex-
ample, the theoremrev (rev l) = l is true regardless of the type of the listl . As
in ML, polymorphism involves type variables: the variablel has type’a . The rewriter
can apply this theorem to simplify a term such asrev (rev [True]) , automatically
matching’a to the typebool andl to the term[True].

In traditional polymorphism, a type variable ranges over all possible types. Isabelle
requires a more refined treatment of polymorphism because certain types are part of
the logical framework and are not available to users. This treatment is based on Wadler
and Blott’stype classes[12], which they introduced to support overloading in the pro-
gramming language Haskell [5]. Nipkow [7] transferred their type system—a clean
combination of overloading and polymorphism—into the context of theorem proving.
Later, Wenzel [13] worked out the logical foundations of axiomatic type classes; he
also produced an elegant implementation.

A type classdenotes a collection of types; asort is a list of type classes and denotes
their intersection. Each type variable has a sort and can be instantiated by any type that

belongs to all of the listed type classes. For example, Isabelle/HOL introduces the type
classzero as follows.

axclass zero ⊆ type

Then, Isabelle/HOL introduces the constant0. (In Isabelle, the symbol:: means “has
type” as well as “has sort.”)

consts "0" :: "’a::zero"

This declaration makes0 polymorphic but only over the type classzero . The two
declarations together definezero to be the class of all types that the constant0 may
have. Next, we can declare a particular type, such asnat , to be a member of class
zero :

instance nat :: zero ..

This declaration informs Isabelle that we intend to use0 as a constant of typenat .
Strictly speaking,0 is not a single constant but a family of constants0:: τ for each
typeτ . These constants can be defined independently of one another.

Isabelle/HOL declares several other constants to be polymorphic over dedicated
type classes. The type classone comprises the types that have the constant1; the
type classesplus andtimes comprise the types that have the infix operators+ and* ,
respectively. Here are the definitions of the type classplus and the infix operator+.

axclass plus ⊆ type
consts "+" :: "[’a::plus, ’a] => ’a" ( infixl 65)

The definitions of the classeszero and times are similar to those presented above.
Now, the sort{zero,one,plus,times} denotes the intersection of the four named
type classes, comprising the types that possess all four of these constants. These will
include the standard arithmetic types, but they may include other types: we have not
yet specified the properties of these constants.

An axiomatic type classis a type class augmented with axioms constraining the
constants. (Wadler and Blott [12] foresaw this possibility.) We can refer to these
axioms in proofs, obtaining theorems specific to the type class. To show that a typeτ

is an instanceof a particular axiomatic type class, we must prove that all the axioms
hold. Typically the axioms refer to overloaded constants, which we define for typeτ

with the objective of satisfying the axioms. Verifying the axioms for typeτ makes all
the theorems proved for the type class immediately available for typeτ . No further
importation or instantiation is necessary.

The implementation burden for axiomatic type classes falls mainly on type infer-
ence. Type checking determines whether or not a polymorphic theorem may be instan-
tiated with a specific type to draw conclusions about specific terms. Isabelle combines
theorems using higher-order unification; type classes complicates the procedure by re-
quiring a treatment of order-sorted type variables. (Before the advent of type classes,
higher-order unification was monomorphic.) Nipkow [7] describes the modifications.
Wenzel [13] proves the soundness of axiomatic type classes via a translation into pure
higher-order logic, but his proof does not reflect the implementation. As just stated, the
mechanism for ensuring that rules are applied correctly is type checking.

Isabelle’s axiomatic type classes provide excellent support for overloading, but they
are not a general abstraction mechanism. Any attempt to use them as such will run into
a number of difficulties:

• Names of constants are required to agree. A specification of monoids that refers
to 0 and+ cannot be used to constrain1 and* .

• A type can be aninstanceof a class in at most one way. This is inconvenient
when there are multiple possibilities, such as whether≤ on lists should denote
the prefix ordering or lexicographic ordering. The same restriction rules out
duality arguments such as reversing the direction of a partial ordering.

Algebraic concepts such as rings and fields are easily defined as axiomatic type
classes. Unfortunately, this does not yield a useful environment for developing ring
theory. The carrier of a ring could only be a type, when realistic examples frequently
require the carrier to be an arbitrary set. Type instance declarations cannot appear in
the middle of a proof, nor can they be rescinded. Type classes for rings and fields
are useful not for formalising abstract algebra but for organising libraries of theorems
about arithmetic types. The type classes for rings and fields are therefore oriented
toward organising arithmetic types and do not convey all the detail found in abstract
algebra. With the definitions given below, a ring is always commutative and has a
multiplicative unit (that is, 1).

Since this paper was first written, Steven Obua has greatly expanded the type class
hierarchy. He has introduced standard terminology—for example, aring need not be
commutative—and added many new concepts. These improvements make the hierar-
chy more general, and Obua has used it to formalize matrices. This paper presents the
original hierarchy in order to avoid complicating the presentation.

3 Axiomatic type classes for arithmetic

This section presents the definitions of the axiomatic type classes used (in Isabelle
2004) to organise the arithmetical theories. They form a hierarchy as new operators
are introduced, such as subtraction, division, and orderings, and as new axioms are
introduced. Because classes can always be combined to form a sort, the hierarchy is a
lattice, with ordered fields at the top and semirings at the bottom.

This hierarchy rests upon existing type classes for orderings. These appear below,
in simplified form. Type classord comprises all types for which the infix relations<

and≤ are defined.1

axclass ord ⊆ type
consts "<" :: "[’a:: ord, ’a] => bool" ( infixl 50)
consts " ≤" :: "[’a:: ord, ’a] => bool" ( infixl 50)

Type classorder comprises the partial orderings, which are reflexive, transitive
and antisymmetric. This type class also specifies the relationship between< and≤.

axclass order ⊆ ord
order_refl: "x ≤ x"
order_trans: "x ≤ y H⇒ y ≤ z H⇒ x ≤ z"
order_antisym: "x ≤ y H⇒ y ≤ x H⇒ x = y"
order_less_le: "(x < y) = (x ≤ y ∧ x 6= y)"

Type classlinorder further constrains the ordering to be linear. Of course, the
numeric types (other thancomplex ) are linearly ordered.

axclass linorder ⊆ order
linorder_linear: "x ≤ y ∨ y ≤ x"

1To avoid a multiplicity of trivial variants of theorems, Isabelle does not define> and≥.

3.1 The Basic Type Class Hierarchy

A semiring(for our purposes) is an algebraic structure with the constants 0 and 1 and
the operators + and×. The binary operators are commutative and associative and they
satisfy the usual distributive law. Moreover, addition can be cancelled from the left.

axclass semiring ⊆ zero, one, plus, times
add_assoc: "(a + b) + c = a + (b + c)"
add_commute: "a + b = b + a"
add_0: "0 + a = a"
add_left_imp_eq: "a + b = a + c H⇒ b=c"

mult_assoc: "(a * b) * c = a * (b * c)"
mult_commute: "a * b = b * a"
mult_1: "1 * a = a"

left_distrib: "(a + b) * c = a * c + b * c"
zero_neq_one: "0 6= 1"

The first line introduces the classsemiring and places it in the sort hierarchy: it is a
subclass ofzero , one , plus andtimes , and therefore of their intersection. The class
is further constrained by axioms.

A ring extends a semiring with unary minus and subtraction, which are related to
addition in the obvious way.

axclass ring ⊆ semiring, minus
left_minus: "- a + a = 0"
diff_minus: "a - b = a + (-b)"

A field extends a ring with a multiplicative inverse (reciprocal) and division, which
are related to multiplication in the obvious way.

axclass field ⊆ ring, inverse
left_inverse: "a 6= 0 H⇒ inverse a * a = 1"
divide_inverse: "a / b = a * inverse b"

Next come ordered versions of these algebraic structures. The basic relation symbol
is≤, but the strict less than relation (<) is also specified. Now, we can define anordered
semiringto be a semiring that is also a linear ordering; it satisfies three further axioms
asserting that 0< 1 and that addition and multiplication preserve the ordering.

axclass ordered_semiring ⊆ semiring, linorder
zero_less_one: "0 < 1"
add_left_mono: "a ≤ b H⇒ c + a ≤ c + b"
mult_strict_left_mono: "a < b H⇒ 0 < c H⇒ c * a < c * b"

An ordered ring is both an ordered semiring and a ring. It must also have the
absolute value function (abs ) defined according to the axiom below. Note thatabs a

may also be written|a|.

axclass ordered_ring ⊆ ordered_semiring, ring
abs_if: " |a| = (if a < 0 then -a else a)"

An ordered fieldis any field that is also an ordered ring.

axclass ordered_field ⊆ ordered_ring, field

An alternative treatment of absolute value would be to define the functionabs once
and for all, using the equation shown above. The definition could be made before types
such asnat and int had been declared and before constants such as0 and < pos-
sessed any definitions. Theorems such as" |a+b | ≤ |a| + |b|" could then be proved
abstractly, from type class axioms. This approach would be fine for numeric types. Un-
fortunately, it presupposes a linear ordering, which is not compatible with other uses
of absolute value in Isabelle. Constraining the functionabs by a type class axiom, as
done here, is more flexible.

3.2 Avoiding Redundant Axioms by Type Class Inclusions

We can refine the definitions given above. One of the semiring axioms becomes redun-
dant when we move to rings: the cancellation of addition can be proved once we have
subtraction. Also, the axiom 0< 1 of ordered semirings is no longer necessary for or-
dered rings: it can be proved from the axiom 06= 1 and the sign laws of multiplication.
Redundant axioms are inelegant, and if they accumulate, they makeinstancedeclara-
tions needlessly long. The type class system allows us to avoid introducing axioms that
will become redundant. The method is best demonstrated using an example. We aban-
don the declarations given above and instead declare a type classalmost_semiring ,
which includes all the axioms except cancellation of addition.

axclass almost_semiring ⊆ zero, one, plus, times
add_assoc: "(a + b) + c = a + (b + c)"
add_commute: "a + b = b + a"
add_0: "0 + a = a"

mult_assoc: "(a * b) * c = a * (b * c)"
mult_commute: "a * b = b * a"
mult_1: "1 * a = a"

left_distrib: "(a + b) * c = a * c + b * c"
zero_neq_one: "0 6= 1"

We extend this type class in two different ways. By introducing subtraction, we obtain
rings, with no redundant axiom about cancellation of addition.

axclass ring ⊆ almost_semiring, minus
left_minus: "- a + a = 0"
diff_minus: "a - b = a + (-b)"

By assuming the axiom for cancellation of addition, we obtain semirings.

axclass semiring ⊆ almost_semiring
add_left_imp_eq: "a + b = a + c H⇒ b=c"

To complete the development, we must prove that all rings are semirings, since it is no
longer true by construction. We do so as follows.

instance ring ⊆ semiring
proof

fix a b c :: ’a
assume "a + b = a + c"
hence "-a + a + b = -a + a + c"

by (simp only: add_assoc)
thus "b = c" by simp

qed

Isabelle demands a proof of the axiom for cancellation of addition. After this is pro-
vided, Isabelle will regard all types that belong to classring to be elements of class
semiring . Proofs of claims are given through the keywordby. After this keyword
comes a command, such as an invocation of the simplifier (simp ), which here refers to
the axiomadd_assoc .

The redundant axiom 0< 1 is handled similarly. We declare a new type class,
almost_ordered_semiring .

axclass almost_ordered_semiring ⊆ semiring, linorder
add_left_mono: "a ≤ b H⇒ c + a ≤ c + b"
mult_strict_left_mono: "a < b H⇒ 0 < c H⇒ c * a < c * b"

Note that it extendssemiring rather thanalmost_semiring . Its purpose is to ensure
that 0< 1 is assumed forordered_semiring but not forordered_ring . Then, we
prove that every ordered ring is an ordered semiring by showing that it satisfies 0< 1.

instance ordered_ring ⊆ ordered_semiring
proof

have "(0::’a) ≤ 1*1" by (rule zero_le_square)
thus "(0::’a) < 1" by (simp add: order_le_less)

qed

Here, the commandby (rule zero_le_square) is a reference to a previously-
proved theorem about ordered rings. This proof of 0< 1 uses 0≤ 1 × 1 as a lemma.

Fields and ordered fields are then defined as before. Fig. 1 is a diagram of the
hierarchy of type classes. The solid lines show inclusions that hold by construction,
while the dashed lines show inclusions that have been proved.

almost_semiring

almost_ordered_semiring

semiring

ring

ordered_semiring

ordered_ring

ordered_field

field

linorder

Figure 1: Type Classes for Organizing Arithmetic Theories

3.3 Reasoning with Type Class Axioms

Now that we have defined some axiomatic type classes, let us use them. Any proof
may refer to the axioms of a type class. Such axioms are entirely different from axioms

asserted at top level: the latter must be taken on faith, while the former will be proved
later in instancedeclarations. If the axioms of a type class are inconsistent, then no
type can be aninstanceof it, but no other harm is done. Isabelle keeps track of the-
orems whose proofs depend on type classes that are not known to contain any types.
Isabelle prints a warning when such theorems are proved and does not permit their use
in justifying otherinstancedeclarations.

Here is a trivial example of a proof. The type classalmost_semiring has an
axiom that says that 0 is the left identity of addition. By the commutative law, it is also
a right identity. Here is the full proof text.

lemma add_0_right: "a + 0 = (a::’a::almost_semiring)"
proof -

have "a + 0 = 0 + a" by (simp only: add_commute)
also have "... = a" by simp
finally show ?thesis .

qed

In the top line is a reference to a type variable, namely’a , which is given an explicit
class, namelyalmost_semiring . All the type classes that the proof requires must be
mentioned in this way; a type variable normally does not belong to any special type
classes, preventing the use of type class axioms. The given theorem statement allows
the use of the axioms ofalmost_semiring . The resulting theorem is specific to this
class and to its descendants, such asring andordered_field . Inclusion for type
classes superficially resembles subtyping. However, it is a deeper concept that reflects
mathematical reasoning in its full generality rather than just the subset relation.

The full theory of rings, fields, etc., comprises approximately 250 theorems. They
are general laws concerning 0, 1,+, −, ×, /, ≤ and<. To maximize generality, each
law refers to the weakest type classes possible. For example, any theorem involving
subtraction requires at least a ring, but if it also involves the ordering then it requires an
ordered ring. Very occasionally, a proof requires more axioms than can be seen from
the theorem statement alone. Here is an example, the lawa × b = 0 ⇐⇒ a = 0 ∨

b = 0.

lemma mult_eq_0_iff: "(a*b = (0::’a::ordered_ring)) = (a=0 ∨ b=0)"

The proof, which is omitted, appeals to the ordering. The factors are either positive,
negative or zero; the non-zero cases contradict the monotonicity of multiplication. This
property of multiplication fails in the ring of integers modulo 4, where 2×2 = 0; it is a
theorem only in ordered rings, even though the theorem statement does not refer to an
ordering. This property of multiplication also holds for fields, when the availability of
division eliminates the need for an ordering. Formally, these are two different theorems.

lemma field_mult_eq_0_iff: "(a*b = (0::’a::field)) = (a=0 ∨ b=0)"

3.4 Instantiating the Type Classes to Specific Types

Now, let us apply the definitions of type classes. Suppose that we have already defined
the type of natural numbers (nat ) and the relevant operators (0, 1, addition, multi-
plication) and have proved their basic properties. Then we may make the following
declaration.

instance nat :: semiring
proof

fix i j k :: nat
show "(i+j)+k = i+(j+k)"

by (rule nat_add_assoc)
show "i+j = j+i"

by (rule nat_add_commute)
show "0+i = i"

by simp
show "(i*j)*k = i*(j*k)"

by (rule nat_mult_assoc)
show "i*j = j*i"

by (rule nat_mult_commute)
show "1*i = i"

by simp
show "(i+j) * k = i*k + j*k"

by (simp add: add_mult_distrib)
show "0 6= (1::nat)"

by simp
assume "k+i = k+j" thus "i=j" by simp

qed

Here we assert that typenat belongs to the type classsemiring . Isabelle checks
what it already knows about this type and then asks us to prove all the axioms given in
the definition of classesalmost_semiring andsemiring . In this example, the ax-
ioms are verified by reference to theorems already been proved for the natural numbers,
such asnat_add_assoc . We could instead have proved each axiom from first princi-
ples. Such details are not important. The essential points are that Isabelle keeps track
of (1) which axioms must be proved and (2) which theorems now become available
for type nat . The first of the two theorems proved above,add_0_right , becomes
available because it holds for all semirings. The second,mult_eq_0_iff , does not:
it requires an ordered ring. This property of multiplication can still be proved from
definitions specific to typenat .

Now, we can prove that the natural numbers are anorderedsemiring. This step
requires proving that typenat is linearly ordered (for a suitable definition of≤) and
that addition and multiplication respect this ordering. If we have already proved that the
type belongs to classessemiring and linorder , then only three additional axioms
will require proofs.

instance nat :: ordered_semiring
proof

fix i j k :: nat
show "0 < (1::nat)"

by simp
show "i ≤ j H⇒ k+i ≤ k+j"

by simp
show "i<j H⇒ 0<k H⇒ k*i < k*j"

by (simp add: mult_less_mono2)
qed

The treatment of other types is similar. Typeint is shown to be a ring as follows.

instance int :: ring
proof

fix i j k :: int
show "(i+j)+k = i+(j+k)"

by (simp add: zadd_assoc)
show "i+j = j+i"

by (simp add: zadd_commute)
show "0+i = i"

by simp
show "- i + i = 0"

by simp
show "i - j = i + (-j)"

by (simp add: zdiff_def)
show "(i*j)*k = i*(j*k)"

by (rule zmult_assoc)
show "i*j = j*i"

by (rule zmult_commute)
show "1*i = i"

by simp
show "(i+j) * k = i*k + j*k"

by (simp add: int_distrib)
show "0 6= (1::int)"

by (simp only: Zero_int_def One_int_def One_nat_def int_int_eq)
qed

Here Isabelle requires proofs of the axioms for the type classesalmost_semiring ,
semiring andring . We could have split thisinstancedeclaration into smaller parts,
but there is no reason to do so. It is good to separate the declarations for the type
classesring andordered_ring : the theorems forring can be used to help develop
the theory of≤ for integers.

Our concept of ordered ring requires a definition of the absolute value function;
recall the discussion at the end of §3.1. Isabelle specifies the functionabs to be over-
loaded for all types that admit unary minus. (The type classminus governs the con-
stants for unary minus, subtraction and absolute value. We could have defined separate
type classes for each of these constants.) Now, as specified by the axiom given in
the definition of an ordered ring, we must define the absolute value function for the
integers.

zabs_def: " |i::int | == if i < 0 then -i else i"

Given this definition, we can show typeint to be an ordered ring.

instance int :: ordered_ring
proof

fix i j k :: int
show "i ≤ j H⇒ k+i ≤ k+j"

by (rule zadd_left_mono)
show "i<j H⇒ 0< H⇒ k*i < k*j"

by (rule zmult_zless_mono2)
show " |i | = (if i<0 then -i else i)"

by (simp only: zabs_def)
qed

The proof of the absolute value axiom merely checks that it has been defined in the
specified manner.

3.5 A Type Class for Powers

All arithmetic types allow a number to be raised to a non-negative power. This ex-
ponentiation operation is meaningful for any type that possesses the number 1 and
multiplication, and therefore, for any semiring. We could include exponentiation in the
definition of a semiring, just asabs is included in the definition of an ordered ring.
However, it is more modular to introduce many simple type classes rather than a few
complicated ones. We can always combine type classes to form sorts.

axclass recpower ⊆ semiring, power
power_0: "a ^ 0 = 1"
power_Suc: "a ^ (Suc n) = a * (a ^ n)"

In this declaration, the type classpower simply denotes the class of all types for which
the power function may be used, without constraining its definition. Non-arithmetic
types may use the power operator differently: to denote then-fold composition of a
relation, for example. Because relational composition does not satisfy basic algebraic
properties such as those of a semiring, no single polymorphic definition of powers
can suffice: this operator must be overloaded. The new class,recpower , extends the
classessemiring and power with two axioms representing the primitive recursive
definition of exponentiation. Many well-known results can be proved in this abstract
setting.

lemma power_one: "1^n = (1::’a::recpower)"
lemma power_one_right: "(a::’a::recpower) ^ 1 = a"
lemma power_add: "(a::’a::recpower) ^ (m+n) = (a^m) * (a^n)"
lemma power_mult: "(a::’a::recpower) ^ (m*n) = (a^m) ^ n"
lemma power_mult_distrib:

"((a::’a::recpower)*b) ^n = (a^n)*(b^n)"

In order to instantiate this type class, we need only make an equivalent definition
for the type of interest. Let us define exponentiation for typenat :

primrec (power)
"p ^ 0 = 1"
"p ^ (Suc n) = (p::nat) * (p ^ n)"

Now, we can demonstrate thatnat belongs to classrecpower .

instance nat :: recpower
proof

fix z n :: nat
show "z^0 = 1" by simp
show "z^(Suc n) = z * (z^n)" by simp

qed

Lemmas such aspower_add now become available for typenat .
Another benefit of the type class approach is uniform simplification. Any lem-

mas can be declared as default simplification rules, which makes them automatically
available to all simplifier invocations. Isabelle/HOL declares many general arithmetic
lemmas, such aspower_one_right andmult_eq_0_iff , as default simplification
rules. Rewrites such asa × b = 0 ⇐⇒ a = 0∨ b = 0 anda1

= a will be performed
for all types in the appropriate type classes, ensuring that simplification behaves simi-
larly for all the arithmetic types.

4 The Treatment of Numeric Literals

Some proofs require the use of literal constants such as 1024, which obviously must
not be expanded to unary notation. In Isabelle, literal constants abbreviate terms of
a data structure that corresponds to two’s complement binary arithmetic. Operations
such as addition, subtraction, multiplication, division, and comparisons are performed
by rewriting on this data structure. Importantly, all this takes place within the logic:
these computations are not hard-wired into Isabelle. They are reasonably efficient: the
computation 1359× −2468= −3354012 takes only a tenth of a second.

Bit strings are defined to be a recursive datatype:

datatype
bin = Pls

| Min
| Bit bin bool ( infixl "BIT" 90)

The constructorpls denotes the value 0, while the constructormin denotes the value
−1; these both terminate the recursion. The third, infix, constructor extends a binary
number with the least significant bit, coded asTrue = 1 andFalse = 0. Addi-
tion, negation, and multiplication can be defined straightforwardly in terms of this data
structure. The two’s complement representation avoids ugly case analyses on the signs
of operands.

The type classnumber denotes all types for which numerals can be defined. The
overloaded functionnumber_of maps bit strings to values of some particular type of
that class.

axclass number ⊆ type
consts number_of :: "bin ⇒ ’a::number"

These definitions let us make separate definitions ofnumber_of for each numeric
type. (If typenat did not require special treatment, then one polymorphic definition of
number_of would have worked for all types.) Each definition requires a separate proof
that the binary arithmetic operations—defined by primitive recursion on bit strings—
agree with numerical arithmetic. Once again, type classes allow us to avoid repeating
the proofs for all the arithmetic types. A new type class specifies just one function,
number_of . All of the agreement proofs refer to the definition of this valuation func-
tion.

Because the binary representation includes negative numbers, we cannot use semir-
ings. The type class for binary arithmetic is based on rings:

axclass number_ring ⊆ number, ring
number_of_Pls: "number_of bin.Pls = 0"
number_of_Min: "number_of bin.Min = - 1"
number_of_BIT: "number_of(w BIT x) = (if x then 1 else 0) +

number_of w + number_of w"

Now, in order to install literal constants for any type that is at least a ring, it is
only necessary to define the functionnumber_of using exactly the recursion specified
above. It remains to set up literal constants for the natural numbers, which form only
a semiring. We cannot prove that typenat belongs to classnumber_ring —which is
why it requires special treatment—but we can enter the type into the trivial class type
number .

instance nat :: number ..

This declaration allows us to define the constantnumber_of for typenat . To evaluate
a bit string as a natural number, we begin by evaluating it as an integer. Then, we apply
the coercion functionnat , which maps negative integers to zero and the others to the
corresponding natural numbers. We arrive at the following definition:

nat_number_of_def:
"(number_of :: bin ⇒ nat) v ==

nat ((number_of :: bin ⇒ int) v)"

Theorems expressing agreement between the binary arithmetic operations and true nat-
ural number arithmetic are proved by case analysis on the signs of the underlying inte-
gers. Notice thatnumber_of for typenat is defined in terms ofnumber_of for type
int . This situation is common with overloading. For instance, the complex number
zero is defined in terms of the real number zero.

The final step in the implementation of new numeric type is to install Isabelle’s
decision procedure for linear arithmetic. This procedure works by deduction within
the logic, just as constant arithmetic does, using theorems to simplify terms and to
deduce contradictions involving inequalities. Using axiomatic type classes, nearly all
of the necessary theorems must be proved only once. This minimises the amount of
ML code needed to set up the procedure for a new arithmetic type.

5 Example: The Natural Numbers with Infinity

Let us extend the type of natural numbers with a new element, infinity. The natural
numbers are extended with∞ by obvious equations such as∞ + n = ∞ × n = ∞.
This type is used in the HOL system for reasoning about the lengths of finite and infinite
lists. Michael Gordon suggested this example, adding “It is an overloading nightmare
in HOL, with lots of theorems covering the various cases when arguments are finite
and infinite. . . .”

Let us define this type in Isabelle and attempt to civilise its theory using type
classes. We declare typenatinf to consist of natural numbers of the formFinite n
and the constantInfinity .

datatype natinf = Finite nat | Infinity

We declare the type to belong to the trivial classeszero , one , plus andtimes .

instance natinf :: "{zero,one,plus,times}" ..

The definitions of zero and one refer to the corresponding natural numbers.

zero_def: "0 == Finite 0"
one_def: "1 == Finite 1"

The definitions of addition and multiplication are by case analysis on whether the
operands are finite or not.

add_def:
"M + N == (case M of Finite m ⇒

(case N of Finite n ⇒ Finite (m+n)
| Infinity ⇒ Infinity)

| Infinity ⇒ Infinity)"

mult_def:

"M * N == (case M of Finite m ⇒

(case N of Finite n ⇒ Finite (m*n)
| Infinity ⇒ Infinity)

| Infinity ⇒ Infinity)"

Typenatinf is an “almost semiring.” The required properties have trivial proofs
by simplification with automatic case splitting.

instance natinf :: almost_semiring
proof

fix M N K :: natinf
show "(M + N) + K = M + (N + K)"

by (simp add: add_def split: natinf.split)
show "M + N = N + M"

by (simp add: add_def split: natinf.split)
show "0 + M = M"

by (simp add: zero_def add_def split: natinf.split)
show "(M * N) * K = M * (N * K)"

by (simp add: mult_def split: natinf.split)
show "M * N = N * M"

by (simp add: mult_def split: natinf.split)
show "1 * M = M"

by (simp add: one_def mult_def split: natinf.split)
show "(M + N) * K = M * K + N * K"

by (simp add: add_def mult_def left_distrib
split: natinf.split)

show "0 6= (1::natinf)"
by (simp add: zero_def one_def)

qed

Now, let us examine the treatment of orderings. We define≤ to extend the anal-
ogous relation on the natural numbers, with∞ obviously the greatest element. We
define< as dictated by the type classes for orderings.

le_def:
"M ≤ N == (case M of Finite m ⇒

(case N of Finite n ⇒ Finite (m ≤n)
| Infinity ⇒ True)

| Infinity ⇒ (N=Infinity))"

less_def: "M < N == M ≤ N ∧ M 6= (N::natinf)"

Typenatinf is linearly ordered. The fivelinorder axioms are stated usingshow
or assume. . . thus. The proofs are again trivial.

instance natinf :: linorder
proof

fix M N K :: natinf
show "M ≤ M" by (simp add: le_def split: natinf.split)
{

assume "M ≤ N" and "N ≤ M" thus "M = N"
by (simp add: le_def split: natinf.split_asm)

next
assume "M ≤ N" and "N ≤ K" thus "M ≤ K"

by (simp add: le_def split: natinf.split_asm)

next
show "(M < N) = (M ≤ N ∧ M 6= N)"

by (simp add: less_def)
show "M ≤ N ∨ N ≤ M"

by (simp add: le_def linorder_linear split: natinf.split)
}

qed

At this point, we have accomplished much by comparison with the equivalent for-
malization in HOL. Type classes (not provided in HOL) allow existing theories to be
re-used. The Isabelle proof script is considerably shorter than the HOL one, but it
provides access to many more theorems. Showing thatnatinf belongs tolinorder

makes Isabelle’s library of facts about linear orderings available. Showing thatnatinf

belongs toalmost_semiring makes a few numeric lemmas available, as well as op-
erators for sums and products over finite sets.

Unfortunately, typenatinf lacks many key properties. Addition cannot be can-
celled and multiplication is not strictly monotonic: we have∞ + M = ∞ = ∞ + N
and∞× M = ∞ = ∞× N for all M andN. The type cannot be shown to be a semir-
ing, and most arithmetic laws require at least this. Without cancellation of addition, we
cannot even prove 0× a = 0, and indeed 0× ∞ = ∞ for typenatinf . We now see
that simply adding infinity to the natural numbers does not yield a theory that satisfies
the usual arithmetic laws. Axiomatic type classes not only allow proofs to be re-used,
but also provide a framework for analysing proposed arithmetic types.

nat

int

rat

real

complex

hypernat

hyperreal

hypercomplex

Figure 2: The Arithmetic Types of Isabelle/HOL

6 Conclusions

Axiomatic type classes work well in practice. Isabelle/HOL is distributed with eight
arithmetic types (Fig. 2). Five of these are traditional: the natural, integer, rational, real
and complex numbers. Three more arise from non-standard analysis: the hypernatural,
hyperreal and hypercomplex numbers. The hyperreal numbers include infinitesimal
and infinite values in addition to the usual real numbers; they allow a rigorous formali-
sation of intuitive arguments about two values being infinitesimally close together, for

example. The hypercomplex numbers are related to the hyperreal numbers in the ob-
vious way. The hypernatural numbers, which are less well known, extend the natural
numbers with infinite values; they form an ordered semiring, and are therefore much
better behaved than typenatinf . All three “hyper” types were defined from the stan-
dard types (using ultrafilters) by Fleuriot [2]. Most users will not require non-standard
analysis, but it is significant that type classes can cope with so many different types.

Subtyping is not the way to formalize such a complicated type hierarchy. Its inel-
egance would become glaringly obvious if we had to derive properties of the natural
numbers from those of the hypercomplex numbers. Users would have to know about
obscure types such as the hypercomplexes and hyperreals simply because of their po-
sition near the top of the hierarchy.

This paper can be seen as advocacy for axiomatic type classes. However, the key
point is simply axiomatic development of abstract mathematics followed by applica-
tion to concrete instances. A theory interpretation construct such as that of IMPS [1]
may also work as a basis for organising theories of arithmetic types. However, in order
to provide a uniform notation, the approach also depends upon overloading. Theory in-
terpretation of itself does not necessarily allow the symbol+ to take on several different
meanings in a single expression.

Also relevant is the constructive algebraic hierarchy formalized in Coq by Geuvers
et al. [3]. They define concepts such as groups and rings. There appears to be no over-
loading of the familiar arithmetic operators. Instead, the work is a basis for developing
constructive algebra.

Acknowledgements. Tobias Nipkow and Markus Wenzel equipped Isabelle with ax-
iomatic type classes. The type classes for ordered rings and fields are based on earlier

